
Introduction 
A technique that SAS programmers often need and find useful is the ability to identify the beginning and ending observation in 
a by-group. The DATA step creates two temporary variables for each BY variable: FIRST.variable and LAST.variable. Performing 
by-group processing to identify FIRST., LAST., and BETWEEN observations is a popular technique with SAS users. Unfortunately, 
PROC SQL users had no way to emulate this very important programming technique. So, after considerable research I came up 
with coding techniques that can be used to emulate these stalwart DATA step programming techniques using subqueries. The 
SQL procedure can be used to correctly emulate and identify FIRST, LAST, and BETWEEN rows in By-Groups, and LAG and LEAD 
nearest neighbor processing. 
 

Example Tables 
The examples used in this paper utilize a database of two tables. (A relational database is a collection of tables.) The data used 
in all the examples in this paper consists of a selection of movies that I’ve viewed over the years. The Movies table consists of 
six columns: title, length, category, year, studio, and rating. Title, category, studio, and rating are defined as character columns 
with length and year being defined as numeric columns. The Actors table consists of three columns: title, actor_leading, and 
actor_supporting, all of which are defined as character columns. The two tables are illustrated below. 
 
MOVIES Table 
 

 
 
ACTORS Table 

Identifying FIRST Rows in By-groups 
 
SQL Code 
 

/*****************************************************/ 

/** ROUTINE.....: FIRST-BY-GROUP-ROWS               **/ 

/** PURPOSE.....: Derive the first (min) row within **/ 

/**               each by-group using a subquery.   **/ 

/*****************************************************/ 

proc sql; 

  create table first_bygroup_rows as 

    select rating, 

           title, 

           'FirstRow' as ByGroup 

      from movies M1 

        where title = 

               (select min(title) 

                  from movies M2 

                    where M1.rating = M2.rating) 

          order by rating, title; 

 

FIRST Rows Results 
 

 
 
 

Identifying LAST Rows in By-groups 
 
SQL Code 
 

/*****************************************************/ 

/** ROUTINE.....: LAST-BY-GROUP-ROWS                **/ 

/** PURPOSE.....: Derive the last (max) row within  **/ 

/**               each by-group using a subquery.   **/ 

/*****************************************************/ 

  create table last_bygroup_rows as 

    select rating, 

           title, 

           'LastRow' as ByGroup 

      from movies M1 

        where title = 

               (select max(title) 

                  from movies M2 

                    where M1.rating = M2.rating) 

          order by rating, title; 

 

LAST Rows Results 

Identifying BETWEEN Rows in By-groups 
 
SQL Code 
 

/************************************************************/ 

/** ROUTINE.....: BETWEEN-BY-GROUP-ROWS                    **/ 

/** PURPOSE.....: Derive not the first (min) row and not   **/ 

/**               the last (max) row within each By-group. **/ 

/************************************************************/ 

 create table between_bygroup_rows as 

    select rating, 

           title, 

           min(title) as Min_Title, 

           max(title) as Max_Title, 

           'BetweenRow' as ByGroup 

      from movies 

        group by rating 

          having CALCULATED min_Title NOT = CALCULATED max_Title 

             AND CALCULATED min_Title NOT = Title 

             AND CALCULATED max_Title NOT = Title 

            order by rating, title; 

 
 
BETWEEN Rows Results 

SQL Code 
 

/***********************************************************/ 

/** ROUTINE.....: CONCATENATE-FIRST-BETWEEN-LAST          **/ 

/** PURPOSE.....: Concatenate the results from the first  **/ 

/**               (min) row, between rows, and last (max) **/ 

/**               row within each by-group, and print.    **/ 

/***********************************************************/ 

  create table first_between_last_rows as 

    select rating, title, bygroup 

      from first_bygroup_rows 

    UNION ALL 

    select rating, title, bygroup 

      from between_bygroup_rows 

    UNION ALL 

    select rating, title, bygroup 

      from last_bygroup_rows; 

  select * from first_between_last_rows; 

quit; 

 
 
FIRST, LAST and BETWEEN Row Results 

Performing Nearest Neighbor Processing by Emulating LAG and LEAD Functionality 
As a general rule SQL queries are designed to perform operations on a row-by-row basis. But, sometimes a problem comes 

along where row-by-row processing will not work. For example, nearest neighbor problems identify content that is close to 

another value. In the DATA step, LAG and LEAD functions are often used to perform these types of operations. But, PROC SQL 

does not support the use of the LAG and LEAD functions, and produces ERROR messages as is shown, below. 

 

SQL Code 

 

PROC SQL NONUMBER ; 

  SELECT Title, Rating, 

         LAG(Title)  AS Lag_Title, 

         LEAD(Title) AS Lead_Title 

    FROM Movies ; 

QUIT ; 

 

SQL Log 

 

PROC SQL NONUMBER ; 

  SELECT Title, Rating, 

         LAG(Title)  AS Lag_Title, 

         LEAD(Title) AS Lead_Title 

    FROM Movies ; 

ERROR: The LAG function is not supported in PROC SQL, it is only valid 

       within the DATA step. 

ERROR: Function LEAD could not be located.   

QUIT ; 

 

 

To overcome the unavailability of the LAG and LEAD functions in PROC SQL, the following code emulates LAG and LEAD 

functionality. 

 

SQL Code 

 

ODS OUTPUT SQL_RESULTS=Movies_with_Row_Numbers ; 

 

PROC SQL NUMBER NOPRINT ; 

 SELECT title, rating FROM mydata.Movies ; 

QUIT ; 

 

PROC SQL NONUMBER ; 

 CREATE TABLE Nearest_Neighbor(drop=Row) as 

   SELECT *, 

     (SELECT Title 

        FROM Movies_with_Row_Numbers 

          WHERE Row = M.Row - 1) AS Previous_Title, 

     (SELECT Title 

        FROM Movies_with_Row_Numbers 

          WHERE Row = M.Row + 1) AS Next_Title 

     FROM Movies_with_Row_Numbers M ; 

 SELECT * FROM Nearest_Neighbor ; 

QUIT ; 

Results 

 

 
 

 

Conclusion 
The SQL procedure is a wonderful tool for SAS users to explore and use in a variety of application situations. This paper has 
presented code that emulates and correctly identifies the FIRST, LAST, and BETWEEN rows in By-Groups using and LAG and 
LEAD nearest neighbor processing using SAS-SQL (PROC SQL). 

